/*
Susan Persky
Proposal title: Emotion and Response to Physician-Provided Genetic Weight Management Messages


HYPOTHESES

Stated-Hyp1: "We hypothesize that anger and fear will have opposite effects on perceptions arising from a clinical encounter. Specifically, we anticipate that anger will trigger increased perceptions of personal control, more heuristic information processing, and decreased cooperation, trust, and risk perception, compared to the control condition, whereas fear will have the opposite effect." (p. 3)

	Test-Hyp1: Perceptions of personal control will be higher in anger vs. control condition.

	Test-HypX: heuristic information processing will be higher in anger vs. control condition.
	// can't test. See note.
	
	Test-Hyp2: Cooperation will be lower in anger vs. control condition.
	// not sure exactly how author planned to measure this, but it seems like behavior change intentions.

	Test-Hyp3: Trust will be lower in anger vs. control condition.
	
	Test-Hyp4: Risk perception will be lower in anger vs. control condition.	
	
	Test-Hyp5: Perceptions of personal control will be lower in fear vs. control condition.

	Test-HypX: heuristic information processing will be lower in fear vs. control condition.
	// can't test. See note.
	
	Test-Hyp6: Cooperation will be higher in fear vs. control condition.
	// not sure exactly how author planned to measure this, but it seems like behavior change intentions.

	Test-Hyp7: Trust will be higher fear vs. control condition.
	
	Test-Hyp8: Risk perception will be higher in fear vs. control condition.	
		

Stated-Hyp2: "genetic information is expected to heighten some effects of emotional state and attenuate others" (p. 3)
NOTE: detailed hypotheses about moderation are on pages 7-8 :

a. The uncontrollable nature of genetic causes are expected to attenuate increases in self-efficacy for those who are in the anger/genetic information condition
b: ...The personalized risk associated with genetic information is expected to be magnified by the influence of fear in heightening risk perception.
c: ...Personalized risk associated with genetic information is expected to increase behavior change intentions for the control and anger conditions. Reduced self-efficacy associated with fear, however, is expected to attenuate this response resulting in the same level of behavior change intention between the genetic and behavior information conditions for those who are in the fear group.

	Test-Hyp9: Genetic information will moderate [lower] effect of anger (vs. control) on perceptions of personal control.

	Test-Hyp10: Genetic information will moderate [increase] effect of fear (vs. control) on risk perception.

	Test-Hyp11: Genetic information will NOT moderate effect of anger (vs. control) on behavior change.
	
	Test-Hyp12: Genetic information will moderate [decrease] effect of fear (vs. control) on behavior change.	
	
Stated-Hyp3: "d: Genetic causal attributions are expected to decrease with anger (via reduced trust in the doctor and his message) and increase with fear (via increased trust in the doctor and his message). However, this is only posited to happen in the genetic information condition." (p. 8)

	Test-Hyp13: Within genetic information condition, genetic causal attributions will decrease with anger vs. control condition.

	Test-Hyp14: Within genetic information condition, genetic causal attributions will increase with fear vs. control condition.
	
	Test-Hyp15: Within genetic information condition, trust in the doctor will mediate the effect of anger on genetic causal attributions.

	Test-Hyp16: Within genetic information condition, trust in the doctor will mediate the effect of fear on genetic causal attributions.

Stated-Hyp4: "e: Information processing and memory is expect to be more heuristic, poorer, and more verbatim for anger, more systematic, detailed and gist-based for fear. We expect the novelty of the genetics information to attenuate decreases in memory among those in the anger group." (p. 8)

	// Can't test. See note. 
	
Stated-Hyp5: "f: Weight stigma perceptions are expected to be increased for anger (via decreased cooperation, reduced expectation for fair treatment, and heuristic/stereotypical processing), and unchanged for fear (given no change in cooperation or expectation for fear treatment). We expect the uncontrollable nature of the genetics message will attenuate perceived stigmatization." (p. 8)
	// see notes.
	
	Test-Hyp17: Percieved stigmatization will be lower in genetic information vs. control condition.
	
	Test-Hyp18: Percieved stigmatization will be higher in anger vs. control condition.
		
	Test-Hyp19: Percieved stigmatization will not differ between fear vs. control condition.
		
	Test-Hyp20: Higher percieved stigmatization in anger vs. control condition will be mediated by cooperation (behavioral intentions).
		
	Test-Hyp21: Higher percieved stigmatization in anger vs. control condition will be mediated by trust.
		

********************************************************************************
NOTES:
- We cannot reasonably measure heuristic processing. It seems from pages 7-8 that 
the author planned to measure this as memory. The only question in the survey 
that pertain to the memory is qualitative and it cannot be easily coded. 

Q22.    Suppose we come back to you in a month.  What are 3 facts or findings 
the doctor shared with you that you are most likely to remember?  You may define 
a "fact or finding" however you wish.

	Fact or finding #1	[    MEDIUM TEXT BOX   ]
	Fact or finding #2  	[    MEDIUM TEXT BOX   ]
	Fact or finding #3 	[    MEDIUM TEXT BOX   ]

We are therefore not testing hypotheses about heuristic processing.

- There is no measure of "fair treatment expectation". We are using trust as a 
proxy. (stated hypothesis 5).	
	*/		

clear all
use "Persky348.dta", clear

********************************************************************************
* MEASURES

* TREATMENT
	* emotion triggered
	tab dov_emotion
	recode dov_emotion (1=1) (3=0) (*=.), gen(treat_fear)
	recode dov_emotion (2=1) (3=0) (*=.), gen(treat_anger)
	
	* genetic information
	tab dov_info
	recode dov_info (1=1) (2=0), gen(treat_geneticinfo)
	tab treat_geneticinfo
	
* OUTCOMES (note: I'm using the outcomes that are stated in the hypothesis)

	* personal control/self-efficacy (p. 10)
		// Q10.   I feel as if I could take the right actions to achieve a healthy weight.
		tab q10
		clonevar control = q10
		replace control=.r if q10==-1
		tab control, mis
			// this was not shown to control condition; so can't test this outcome!
		
	* trust
		// Q13. If this were your doctor, how much would you trust him?
		tab q13
		clonevar trust = q13
		replace trust=.r if q13==-1
		tab trust, mis
		
		
	* risk perception
		// Q8 Overall, how do you think your chances of developing breast cancer in the future compare to the average woman your age?
		tab q8
		clonevar risk = q8
		replace risk=.r if q8==-1
		tab risk, mis

	* behavior change intention/cooperation (see pages 11-12)
		/*
		Q18.   I intend to make changes to my diet in the next 6 months.
		Q19.   I intend to make changes to get more exercise in the next 6 months.
		Q20.   How likely is it you will try to change your diet in the next 6 months?
		Q21.   How likely is it you will try to get more exercise in the next 6 months?
		*/
		foreach var in q18 q19 q20 q21 {
			replace `var'=. if `var'==-1
		}
		gen behaviorchange=q18 +q19+q20+q21
		tab behaviorchange, mis	

	* genetic causal attribution	
	/*
	To what extent do you agree or disagree that each of the following factors cause or contribute to your body weight?
	Q15.   Genetics
	*/
		
		tab q15
		clonevar geneticattribution = q15
		replace geneticattribution=.r if q15==-1
		tab geneticattribution, mis	

	*Weight stigma perceptions (see page 10)
		/*Q11.   This doctor thinks less of me because of my weight.
		Q12.   This doctor believes I won't follow his advice.*/
		foreach var in q11 q12 {
			replace `var'=. if `var'==-1
		}
		gen weightstigma=q11+q12

********************************************************************************	
* ANALYSIS 

*Test-Hyp1: Perceptions of personal control will be higher in anger vs. control condition.
	reg control i.treat_anger
	// reject. .301
	tess 1.treat_anger +, init(Persky348) bonf(5)
	
*Test-Hyp2: Cooperation will be lower in anger vs. control condition.
	reg behaviorchange i.treat_anger
	// reject. .332
	tess 1.treat_anger -, bonf(5)
	
*Test-Hyp3: Trust will be lower in anger vs. control condition.
 	reg trust i.treat_anger
	// reject. .638
	tess 1.treat_anger -, bonf(5)
	
*Test-Hyp4: Risk perception will be lower in anger vs. control condition.	
	reg risk i.treat_anger
	// reject. .655
	tess 1.treat_anger -, bonf(5)
	
*Test-Hyp5: Perceptions of personal control will be lower in fear vs. control condition.
	reg control i.treat_fear
	// reject. 0.094
	tess 1.treat_fear -, bonf(5)
	
*Test-Hyp6: Cooperation will be higher in fear vs. control condition.
	reg behaviorchange i.treat_fear
	// reject. .216
	tess 1.treat_fear +, bonf(5)
	
*Test-Hyp7: Trust will be higher fear vs. control condition.
	reg trust i.treat_fear
	// reject. .215
	tess 1.treat_fear +, bonf(5)
	
*Test-Hyp8: Risk perception will be higher in fear vs. control condition.	
	reg risk i.treat_fear		
	// reject. .519
	tess 1.treat_fear +, bonf(5)
	
*Test-Hyp9: Genetic information will moderate [lower] effect of anger (vs. control) on perceptions of personal control.
	reg control i.treat_anger##i.treat_geneticinfo
	// reject. 0.691
	tess 1.treat_anger#1.treat_geneticinfo -, bonf(2)
	
*Test-Hyp10: Genetic information will moderate [increase] effect of fear (vs. control) on risk perception.
	reg risk i.treat_fear##i.treat_geneticinfo
	// reject. 0.703
	tess 1.treat_fear#1.treat_geneticinfo +, bonf(2)
	
*Test-Hyp11: Genetic information will NOT moderate effect of anger (vs. control) on behavior change.
	reg behaviorchange i.treat_anger##i.treat_geneticinfo
	// do not reject. 0.202
	tess 1.treat_anger#1.treat_geneticinfo, bonf(2)
	
*Test-Hyp12: Genetic information will moderate [decrease] effect of fear (vs. control) on behavior change.	
	reg behaviorchange i.treat_fear##i.treat_geneticinfo
	// reject. 0.265	
	tess 1.treat_fear#1.treat_geneticinfo -, bonf(2)
	
*Test-Hyp13: Within genetic information condition, genetic causal attributions will decrease with anger vs. control condition.
	reg geneticattribution i.treat_anger if treat_geneticinfo==1
	// reject. 0.627
	tess 1.treat_anger -
	
*Test-Hyp14: Within genetic information condition, genetic causal attributions will increase with fear vs. control condition.
	reg geneticattribution i.treat_fear if treat_geneticinfo==1
	// reject. 0.217
	tess 1.treat_fear +
	
*Test-Hyp15: Within genetic information condition, trust in the doctor will mediate the effect of anger on genetic causal attributions.
	reg geneticattribution treat_anger if treat_geneticinfo==1
	est sto m1
	
	reg geneticattribution treat_anger trust if treat_geneticinfo==1
	est sto m2
	
	suest m1 m2 
	test [m1_mean]treat_anger=[m2_mean]treat_anger 
	// reject. 0.178
	
	tess, mediation
	
*Test-Hyp16: Within genetic information condition, trust in the doctor will mediate the effect of fear on genetic causal attributions.	

	reg geneticattribution treat_fear if treat_geneticinfo==1
	est sto m1
	
	reg geneticattribution treat_fear trust if treat_geneticinfo==1
	est sto m2
	
	suest m1 m2 
	test [m1_mean]treat_fear=[m2_mean]treat_fear
	// reject. 0.356

	tess, mediation	
	
*Test-Hyp17: Percieved stigmatization will be lower in genetic information vs. control condition.
	reg weightstigma i.treat_geneticinfo
	// reject. 0.353
	tess 1.treat_geneticinfo -
	
*Test-Hyp18: Percieved stigmatization will be higher in anger vs. control condition.
	reg weightstigma i.treat_anger	
	// reject. 0.470
	tess 1.treat_anger +, bonf(5)	
	
*Test-Hyp19: Percieved stigmatization will NOT differ between fear vs. control condition.
	reg weightstigma i.treat_fear
	// reject. 0.371
	tess 1.treat_fear, bonf(5)
	
*Test-Hyp20: Higher percieved stigmatization in anger vs. control condition will be mediated by cooperation (behavioral intentions).
	reg weightstigma treat_anger
	est sto m1
	
	reg weightstigma treat_anger behaviorchange
	est sto m2
	
	suest m1 m2 
	test [m1_mean]treat_anger=[m2_mean]treat_anger
	// reject. 0.215

	tess, mediation		
	
*Test-Hyp21: Higher percieved stigmatization in anger vs. control condition will be mediated by trust.
	reg weightstigma treat_anger
	est sto m1
	
	reg weightstigma treat_anger trust
	est sto m2
	
	suest m1 m2 
	test [m1_mean]treat_anger=[m2_mean]treat_anger
	// reject. 0.565		

	tess, mediation		
